
Changelog
    

All notable changes to this project will be documented in this file.

  
    
  
  [1.0.0-rc.2] - 2025-06-24



  
    
  
  Added


	Support for multiple cloud storage providers:	Local filesystem (Volume adapter)
	Google Cloud Storage (GCS adapter)
	Amazon S3 (S3 adapter)
	Cloudflare R2 (via S3 adapter with provider: :cloudflare)
	DigitalOcean Spaces (via S3 adapter with provider: :digitalocean)
	Tigris (via S3 adapter with provider: :tigris)


	Multi-cloud support with multiple Cloud modules
	Dynamic cloud configuration for runtime-determined storage
	Signed URL support for secure uploads and downloads
	Phoenix LiveView integration for direct-to-cloud uploads
	Development router for local file uploads/downloads
	Telemetry integration for monitoring and metrics
	Comprehensive test suite with live service testing


  
    
  
  Changed


	Switched from multiple strategies per cloud to one adapter per cloud approach
	Renamed Strategy modules to Adapter modules
	Replaced external dependencies (gcs_signed_url, google_api_storage, goth, ex_aws) with req
	Renamed cloudflare_r2 provider to cloudflare for consistency
	Cloud modules now act as supervisors for adapters that need background processes


  
    
  
  Fixed


	Router configuration to allow cloud module in location config
	Automatic setting of correct GCS options for uploads
	Warning messages for cloud supervisors with no children


  
    
  
  Internal


	Added validate_config callback to adapter behavior
	Improved Location struct with derived inspect that ignores :config
	Enhanced documentation and README with core concepts



  

    
Getting Started
    

Welcome to Buckets! This guide will walk you through setting up Buckets in your Phoenix application.

  
    
  
  Installation


Add buckets to your list of dependencies in mix.exs:
def deps do
  [
    {:buckets, "~> 1.0.0-rc.2"}
  ]
end
Then run:
mix deps.get


  
    
  
  Quick Start



  
    
  
  1. Create a Cloud Module


Create a Cloud module in your application:
defmodule MyApp.Cloud do
  use Buckets.Cloud, otp_app: :my_app
end

  
    
  
  2. Add to Supervision Tree


Add your Cloud module to your application's supervision tree:
def start(_type, _args) do
  children = [
    # ... other children
    MyApp.Cloud
  ]

  Supervisor.start_link(children, opts)
end

  
    
  
  3. Configure Your Adapter


For development with local filesystem:
# config/dev.exs
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.Volume,
  bucket: "tmp/buckets",
  base_url: "http://localhost:4000"
For production with S3:
# config/prod.exs
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  bucket: "my-production-bucket",
  region: "us-east-1",
  access_key_id: System.fetch_env!("AWS_ACCESS_KEY_ID"),
  secret_access_key: System.fetch_env!("AWS_SECRET_ACCESS_KEY")

  
    
  
  4. Basic Usage


Upload a file:
{:ok, object} = MyApp.Cloud.insert("/path/to/file.pdf")
Handle file uploads from a form:
def create(conn, %{"upload" => upload}) do
  {:ok, object} = MyApp.Cloud.insert(Buckets.Object.from_upload(upload))
  # Save object metadata to database
  # ...
end

  
    
  
  Next Steps


	Learn about Core Concepts
	Explore different Storage Adapters
	Set up Direct Uploads with LiveView



  

    
Core Concepts
    

Understanding these core concepts will help you use Buckets effectively.

  
    
  
  Cloud Modules


A Cloud module is your main interface to cloud storage, similar to how an Ecto Repo is your interface to a database. You define Cloud modules in your application:
defmodule MyApp.Cloud do
  use Buckets.Cloud, otp_app: :my_app
end
Each Cloud module:
	Connects to one storage backend at a time
	Can be configured differently per environment
	Provides a consistent API regardless of the storage backend


  
    
  
  Adapters


Adapters implement the actual storage operations for different backends:
	Buckets.Adapters.Volume - Local filesystem (great for development)
	Buckets.Adapters.S3 - Amazon S3 and compatible services
	Buckets.Adapters.GCS - Google Cloud Storage

Each adapter has its own configuration requirements and capabilities.

  
    
  
  Objects


Buckets.Object structs represent files in your system. Objects have:
	UUID - Unique identifier
	Filename - Original filename
	Data - The actual file content (when loaded)
	Metadata - Content type, size, and custom metadata
	Location - Where the file is stored
	Stored? - Whether it's been uploaded to cloud storage


  
    
  
  Object Lifecycle


	Created - Object exists in memory/local disk
object = Buckets.Object.from_file("/tmp/upload.pdf")
# stored?: false, location: NotConfigured

	Stored - Object uploaded to cloud storage
{:ok, stored} = MyApp.Cloud.insert(object)
# stored?: true, location: configured

	Loaded - Object data fetched from cloud
{:ok, loaded} = MyApp.Cloud.load(stored)
{:ok, data} = Buckets.Object.read(loaded)



  
    
  
  Locations


A Buckets.Location represents where an object is stored. It contains:
	Path - The storage path (e.g., "uploads/123/file.pdf")
	Config - Either adapter configuration or a Cloud module reference


  
    
  
  Signed URLs


Signed URLs provide temporary, secure access to objects:
{:ok, signed_url} = MyApp.Cloud.url(object, expires_in: 3600)
# Use signed_url.url for direct access
These are especially useful for:
	Direct browser downloads
	LiveView direct uploads
	Temporary file sharing


  
    
  
  Supervision


Some adapters (like GCS) need background processes for authentication. The Cloud module supervisor handles this automatically when added to your supervision tree.
Adapters that need supervision:
	GCS (for auth token management)

Adapters that don't need supervision:
	Volume
	S3


  
    
  
  Configuration


Configuration can be:

  
    
  
  Static (compile-time)


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  bucket: "my-bucket"

  
    
  
  Dynamic (runtime)


MyApp.Cloud.put_dynamic_config([
  adapter: Buckets.Adapters.S3,
  bucket: "tenant-bucket"
])
This enables multi-tenant applications where each tenant has their own storage configuration.


  

    
Writing Custom Adapters
    

This guide explains how to create your own storage adapter for Buckets.

  
    
  
  Overview


An adapter implements the Buckets.Adapter behaviour to provide storage operations for a specific backend. You might create a custom adapter for:
	Proprietary storage systems
	Specialized cloud providers
	Hybrid storage solutions
	Testing purposes


  
    
  
  Basic Structure


defmodule MyApp.Adapters.CustomStorage do
  @behaviour Buckets.Adapter

  @impl true
  def validate_config(config) do
    # Validate configuration
  end

  @impl true
  def put(object, remote_path, config) do
    # Upload object
  end

  @impl true
  def get(remote_path, config) do
    # Download data
  end

  @impl true
  def delete(remote_path, config) do
    # Delete object
  end

  @impl true
  def url(remote_path, config) do
    # Generate signed URL
  end

  # Optional - only if you need supervised processes
  @impl true
  def child_spec(config, cloud_module) do
    # Return supervisor child specification
  end
end

  
    
  
  Implementing Callbacks



  
    
  
  validate_config/1


Validates and normalizes configuration:
@impl true
def validate_config(config) do
  with {:ok, config} <- Keyword.validate(config, [
         :adapter,
         :bucket,
         :api_key,
         :endpoint,
         :path
       ]),
       {:ok, config} <- Buckets.Adapter.validate_required(config, [
         :bucket,
         :api_key,
         :endpoint
       ]) do
    {:ok, config}
  end
end

  
    
  
  put/3


Uploads an object to storage:
@impl true
def put(%Buckets.Object{} = object, remote_path, config) do
  data = Buckets.Object.read!(object)
  
  case upload_to_storage(remote_path, data, config) do
    :ok -> 
      {:ok, %{}}
    
    {:error, reason} -> 
      {:error, reason}
  end
end

  
    
  
  get/2


Downloads data from storage:
@impl true
def get(remote_path, config) do
  case download_from_storage(remote_path, config) do
    {:ok, data} -> 
      {:ok, data}
      
    {:error, :not_found} -> 
      {:error, :not_found}
      
    {:error, reason} -> 
      {:error, reason}
  end
end

  
    
  
  delete/2


Deletes an object:
@impl true
def delete(remote_path, config) do
  case delete_from_storage(remote_path, config) do
    :ok -> 
      {:ok, %{}}
      
    {:error, reason} -> 
      {:error, reason}
  end
end

  
    
  
  url/2


Generates signed URLs:
@impl true
def url(remote_path, config) do
  expires_in = Keyword.get(config, :expires_in, 3600)
  for_upload = Keyword.get(config, :for_upload, false)
  
  signed_url = generate_signed_url(
    remote_path, 
    expires_in, 
    for_upload,
    config
  )
  
  location = Buckets.Location.new(remote_path, config)
  
  {:ok, %Buckets.SignedURL{
    url: signed_url,
    location: location
  }}
end

  
    
  
  child_spec/2 (Optional)


For adapters needing background processes:
@impl true
def child_spec(config, cloud_module) do
  %{
    id: {__MODULE__, cloud_module},
    start: {MyApp.AuthServer, :start_link, [[
      cloud: cloud_module,
      config: config
    ]]},
    restart: :permanent
  }
end

  
    
  
  Best Practices



  
    
  
  1. Configuration Validation


Always validate configuration thoroughly:
def validate_config(config) do
  # Use Keyword.validate for known options
  with {:ok, config} <- Keyword.validate(config, @known_opts),
       # Check required fields
       {:ok, config} <- validate_required(config, @required_opts),
       # Custom validation
       :ok <- validate_endpoint(config[:endpoint]) do
    {:ok, config}
  end
end

  
    
  
  2. Error Handling


Return consistent error tuples:
# Good
{:error, :not_found}
{:error, :unauthorized}
{:error, {:http_error, 500}}

# Bad
{:error, "not found"}
:error
nil

  
    
  
  3. Metadata Handling


Use object metadata appropriately:
def put(object, path, config) do
  headers = build_headers(object.metadata)
  # Include content-type, cache-control, etc.
end

  
    
  
  4. Telemetry Integration


Emit telemetry events:
def put(object, path, config) do
  metadata = %{
    adapter: __MODULE__,
    path: path,
    size: byte_size(data)
  }
  
  :telemetry.span(
    [:my_adapter, :put],
    metadata,
    fn ->
      result = do_upload(object, path, config)
      {result, metadata}
    end
  )
end

  
    
  
  Testing Your Adapter


Create comprehensive tests:
defmodule MyApp.Adapters.CustomStorageTest do
  use ExUnit.Case
  
  setup do
    config = [
      adapter: MyApp.Adapters.CustomStorage,
      bucket: "test-bucket",
      api_key: "test-key"
    ]
    
    {:ok, config: config}
  end
  
  test "validates configuration", %{config: config} do
    assert {:ok, _} = MyApp.Adapters.CustomStorage.validate_config(config)
    
    invalid = Keyword.delete(config, :bucket)
    assert {:error, [:bucket]} = MyApp.Adapters.CustomStorage.validate_config(invalid)
  end
  
  test "uploads objects", %{config: config} do
    object = Buckets.Object.from_file("test/fixtures/sample.pdf")
    
    assert {:ok, _} = MyApp.Adapters.CustomStorage.put(
      object,
      "test/sample.pdf",
      config
    )
  end
end

  
    
  
  Example: FTP Adapter


Here's a simplified FTP adapter example:
defmodule MyApp.Adapters.FTP do
  @behaviour Buckets.Adapter
  
  @impl true
  def validate_config(config) do
    with {:ok, config} <- Keyword.validate(config, [
           :adapter, :host, :port, :username, 
           :password, :bucket, :path
         ]),
         {:ok, config} <- validate_required(config, [
           :host, :username, :password, :bucket
         ]) do
      config = Keyword.put_new(config, :port, 21)
      {:ok, config}
    end
  end
  
  @impl true
  def put(object, remote_path, config) do
    with {:ok, conn} <- connect(config),
         {:ok, data} <- Buckets.Object.read(object),
         :ok <- ftp_put(conn, remote_path, data) do
      {:ok, %{}}
    end
  end
  
  @impl true
  def get(remote_path, config) do
    with {:ok, conn} <- connect(config),
         {:ok, data} <- ftp_get(conn, remote_path) do
      {:ok, data}
    end
  end
  
  # ... implement other callbacks
end

  
    
  
  Publishing Your Adapter


	Create a separate package
	Name it buckets_[provider] (e.g., buckets_ftp)
	Add buckets as a dependency
	Document configuration options
	Provide usage examples
	Publish to Hex.pm


  
    
  
  Need Help?


	Check existing adapters for examples
	Review the Buckets.Adapter behaviour documentation
	Open an issue for clarification
	Submit a PR to improve this guide!



  

    
File Uploads
    

This guide covers various ways to handle file uploads with Buckets.

  
    
  
  Controller Uploads


The most common scenario is handling file uploads in Phoenix controllers.

  
    
  
  Basic Upload


defmodule MyAppWeb.DocumentController do
  use MyAppWeb, :controller
  
  def create(conn, %{"document" => %{"file" => upload}}) do
    # Create object from Plug.Upload
    object = Buckets.Object.from_upload(upload)
    
    # Upload to cloud storage
    case MyApp.Cloud.insert(object) do
      {:ok, stored_object} ->
        # Save metadata to database
        {:ok, document} = Documents.create_document(%{
          filename: stored_object.filename,
          path: stored_object.location.path,
          content_type: stored_object.metadata.content_type,
          size: stored_object.metadata.content_size
        })
        
        conn
        |> put_flash(:info, "Document uploaded successfully")
        |> redirect(to: ~p"/documents/#{document}")
        
      {:error, reason} ->
        conn
        |> put_flash(:error, "Upload failed: #{inspect(reason)}")
        |> redirect(to: ~p"/documents/new")
    end
  end
end

  
    
  
  Multiple File Upload


def create(conn, %{"files" => uploads}) when is_list(uploads) do
  results = Enum.map(uploads, fn upload ->
    object = Buckets.Object.from_upload(upload)
    MyApp.Cloud.insert(object)
  end)
  
  successful = Enum.filter(results, &match?({:ok, _}, &1))
  failed = Enum.filter(results, &match?({:error, _}, &1))
  
  conn
  |> put_flash(:info, "Uploaded #{length(successful)} files")
  |> maybe_put_error_flash(failed)
  |> redirect(to: ~p"/documents")
end

  
    
  
  Direct File Upload


Upload files directly from disk:
# From a file path
{:ok, object} = MyApp.Cloud.insert("/path/to/file.pdf")

# With custom metadata
object = Buckets.Object.new(
  Ecto.UUID.generate(),
  "report.pdf",
  metadata: %{
    author: "John Doe",
    category: "financial"
  }
)

{:ok, stored} = MyApp.Cloud.insert(object)

  
    
  
  Post-Upload Processing


Process files after upload:
def create(conn, %{"image" => upload}) do
  with object <- Buckets.Object.from_upload(upload),
       {:ok, stored} <- MyApp.Cloud.insert(object),
       {:ok, _} <- generate_thumbnails(stored) do
    # Success
  else
    {:error, reason} -> 
      # Handle error
  end
end

defp generate_thumbnails(object) do
  # Load the image data
  {:ok, loaded} = MyApp.Cloud.load(object)
  {:ok, data} = Buckets.Object.read(loaded)
  
  # Generate thumbnails
  sizes = [
    {:thumb, 150},
    {:medium, 500},
    {:large, 1200}
  ]
  
  Enum.each(sizes, fn {name, width} ->
    resized = resize_image(data, width)
    
    thumb_object = Buckets.Object.new(
      object.uuid,
      "#{name}_#{object.filename}",
      metadata: %{content_type: "image/jpeg"}
    )
    
    MyApp.Cloud.insert(thumb_object)
  end)
end

  
    
  
  Next Steps


	Learn about Direct Uploads with LiveView
	Explore Signed URLs for secure access
	Set up Error Handling strategies



  

    
Direct Uploads with LiveView
    

This guide shows how to implement direct-to-cloud uploads using Phoenix LiveView, allowing users to upload files directly to your storage provider without going through your server.

  
    
  
  Overview


Direct uploads improve performance and reduce server load by:
	Uploading files directly from the browser to cloud storage
	Reducing bandwidth usage on your servers
	Enabling progress tracking and cancellation
	Supporting large file uploads


  
    
  
  Basic Setup



  
    
  
  1. Configure Your Cloud Module


Enable direct uploads by setting the uploader option:
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  bucket: "my-uploads",
  uploader: "S3",  # or "GCS" for Google Cloud Storage
  # ... other config

  
    
  
  2. Basic LiveView Upload


defmodule MyAppWeb.UploadLive do
  use MyAppWeb, :live_view

  @impl true
  def mount(_params, _session, socket) do
    {:ok,
     socket
     |> assign(:uploaded_files, [])
     |> allow_upload(:avatar,
       accept: ~w(.jpg .jpeg .png),
       max_entries: 1,
       max_file_size: 5_000_000,
       external: &presign_upload/2
     )}
  end

  @impl true
  def render(assigns) do
    ~H"""
    <form id="upload-form" phx-submit="save" phx-change="validate">
      <.live_file_input upload={@uploads.avatar} />
      <button type="submit">Upload</button>
    </form>

    <div :for={file <- @uploaded_files}>
      Uploaded: <%= file.filename %>
    </div>
    """
  end

  defp presign_upload(entry, socket) do
    {:ok, config} = MyApp.Cloud.live_upload(entry)
    {:ok, config, socket}
  end

  @impl true
  def handle_event("validate", _params, socket) do
    {:noreply, socket}
  end

  @impl true
  def handle_event("save", _params, socket) do
    uploaded_files =
      consume_uploaded_entries(socket, :avatar, fn %{key: key}, entry ->
        object = Buckets.Object.from_upload({entry, %{key: key}})
        {:ok, object}
      end)

    {:noreply, update(socket, :uploaded_files, &(&1 ++ uploaded_files))}
  end
end

  
    
  
  Multiple File Uploads


def mount(_params, _session, socket) do
  {:ok,
   socket
   |> assign(:uploaded_files, [])
   |> allow_upload(:documents,
     accept: ~w(.pdf .doc .docx),
     max_entries: 10,
     max_file_size: 10_000_000,
     external: &presign_upload/2,
     auto_upload: true
   )}
end

  
    
  
  Error Handling


def render(assigns) do
  ~H"""
  <div :for={err <- upload_errors(@uploads.avatar)} class="error">
    <%= error_to_string(err) %>
  </div>
  """
end

defp error_to_string(:too_large), do: "File is too large"
defp error_to_string(:not_accepted), do: "File type not accepted"
defp error_to_string(:too_many_files), do: "Too many files"
defp error_to_string(:external_client_failure), do: "Upload failed"

defp presign_upload(entry, socket) do
  case MyApp.Cloud.live_upload(entry) do
    {:ok, config} ->
      {:ok, config, socket}
      
    {:error, reason} ->
      {:error, "Failed to generate upload URL: #{inspect(reason)}", socket}
  end
end

  
    
  
  CORS Configuration


For direct uploads to work, configure CORS on your bucket:

  
    
  
  S3 CORS


<?xml version="1.0" encoding="UTF-8"?>
<CORSConfiguration>
  <CORSRule>
    <AllowedOrigin>https://myapp.com</AllowedOrigin>
    <AllowedMethod>PUT</AllowedMethod>
    <AllowedMethod>POST</AllowedMethod>
    <AllowedHeader>*</AllowedHeader>
    <ExposeHeader>ETag</ExposeHeader>
    <MaxAgeSeconds>3000</MaxAgeSeconds>
  </CORSRule>
</CORSConfiguration>

  
    
  
  GCS CORS


[
  {
    "origin": ["https://myapp.com"],
    "method": ["PUT", "POST"],
    "responseHeader": ["*"],
    "maxAgeSeconds": 3600
  }
]


  

    
Signed URLs
    

Signed URLs provide temporary, secure access to objects in cloud storage without requiring authentication. This guide explains how to generate and use signed URLs with Buckets.

  
    
  
  What are Signed URLs?


Signed URLs are time-limited URLs that grant temporary access to private objects. They include:
	The object location
	Expiration time
	Cryptographic signature
	Optional permissions (read/write)


  
    
  
  Basic Usage



  
    
  
  Generating Download URLs


# Generate a signed URL for downloads (default: 1 hour)
{:ok, signed_url} = MyApp.Cloud.url(object)

# Access the URL string
download_url = signed_url.url

# Or use String protocol
download_url = to_string(signed_url)

  
    
  
  Custom Expiration


# Expires in 5 minutes
{:ok, signed_url} = MyApp.Cloud.url(object, expires_in: 300)

# Expires in 24 hours
{:ok, signed_url} = MyApp.Cloud.url(object, expires_in: 86400)

# Expires in 7 days (maximum for S3)
{:ok, signed_url} = MyApp.Cloud.url(object, expires_in: 604800)

  
    
  
  Upload URLs


Generate URLs for direct uploads:
# Create a placeholder object
object = Buckets.Object.new(
  Ecto.UUID.generate(),
  "document.pdf",
  location: {"uploads/documents/document.pdf", MyApp.Cloud}
)

# Generate upload URL
{:ok, upload_url} = MyApp.Cloud.url(object, 
  expires_in: 3600,
  for_upload: true
)

  
    
  
  Use Cases



  
    
  
  Secure File Downloads


defmodule MyAppWeb.DocumentController do
  def download(conn, %{"id" => id}) do
    document = Documents.get!(id)
    
    # Load object from storage location
    object = Buckets.Object.new(
      document.id,
      document.filename,
      location: {document.storage_path, MyApp.Cloud}
    )
    
    # Generate temporary download URL
    {:ok, signed_url} = MyApp.Cloud.url(object, expires_in: 300)
    
    # Redirect to signed URL
    redirect(conn, external: signed_url.url)
  end
end

  
    
  
  Email Attachments


defmodule MyApp.Mailer do
  def invoice_email(user, invoice) do
    # Generate URL valid for 7 days
    {:ok, download_url} = MyApp.Cloud.url(invoice.document, 
      expires_in: 604800
    )
    
    new()
    |> to(user.email)
    |> subject("Your Invoice")
    |> html_body("""
    <p>Your invoice is ready!</p>
    <p><a href="#{download_url}">Download Invoice</a></p>
    <p>This link expires in 7 days.</p>
    """)
    |> deliver()
  end
end

  
    
  
  API Responses


defmodule MyAppWeb.API.FileController do
  def show(conn, %{"id" => id}) do
    file = Files.get!(id)
    
    # Generate short-lived URL
    {:ok, signed_url} = MyApp.Cloud.url(file.object, 
      expires_in: 600  # 10 minutes
    )
    
    json(conn, %{
      id: file.id,
      filename: file.filename,
      size: file.size,
      content_type: file.content_type,
      download_url: signed_url.url,
      expires_at: DateTime.add(DateTime.utc_now(), 600)
    })
  end
end


  

    
Multi-Cloud Setup
    

This guide explains how to use multiple cloud storage providers simultaneously in your application.

  
    
  
  Why Multi-Cloud?


Multi-cloud setups are useful for:
	Geographic distribution - Store files closer to users
	Cost optimization - Use the most cost-effective provider for each use case
	Redundancy - Failover between providers
	Compliance - Meet data residency requirements
	Migration - Gradually move between providers


  
    
  
  Basic Multi-Cloud Setup



  
    
  
  1. Define Multiple Cloud Modules


# For local development
defmodule MyApp.LocalCloud do
  use Buckets.Cloud, otp_app: :my_app
end

# For user uploads
defmodule MyApp.S3Cloud do
  use Buckets.Cloud, otp_app: :my_app
end

# For backups and archives
defmodule MyApp.GCSCloud do
  use Buckets.Cloud, otp_app: :my_app
end

# For CDN assets
defmodule MyApp.R2Cloud do
  use Buckets.Cloud, otp_app: :my_app
end

  
    
  
  2. Configure Each Cloud


# config/dev.exs
config :my_app, MyApp.LocalCloud,
  adapter: Buckets.Adapters.Volume,
  bucket: "tmp/local_storage",
  base_url: "http://localhost:4000"

# config/config.exs or runtime.exs
config :my_app, MyApp.S3Cloud,
  adapter: Buckets.Adapters.S3,
  bucket: System.get_env("S3_BUCKET"),
  region: "us-east-1",
  access_key_id: System.get_env("AWS_ACCESS_KEY_ID"),
  secret_access_key: System.get_env("AWS_SECRET_ACCESS_KEY")

config :my_app, MyApp.GCSCloud,
  adapter: Buckets.Adapters.GCS,
  bucket: System.get_env("GCS_BUCKET"),
  service_account_credentials: System.get_env("GOOGLE_CREDENTIALS")

config :my_app, MyApp.R2Cloud,
  adapter: Buckets.Adapters.S3,
  provider: :cloudflare,
  bucket: System.get_env("R2_BUCKET"),
  endpoint_url: System.get_env("R2_ENDPOINT"),
  access_key_id: System.get_env("R2_ACCESS_KEY"),
  secret_access_key: System.get_env("R2_SECRET_KEY")

  
    
  
  3. Add to Supervision Tree


def start(_type, _args) do
  children = [
    # ... other processes
    MyApp.GCSCloud,  # Only GCS needs supervision
    # S3, R2, and Volume don't need supervision
  ]
  
  Supervisor.start_link(children, opts)
end

  
    
  
  Usage Patterns



  
    
  
  Content-Based Routing


Route files to different clouds based on content:
defmodule MyApp.Storage do
  def store_file(upload) do
    cloud = determine_cloud(upload)
    object = Buckets.Object.from_upload(upload)
    cloud.insert(object)
  end
  
  defp determine_cloud(upload) do
    case upload.content_type do
      "image/" <> _ -> MyApp.R2Cloud      # Images to CDN
      "video/" <> _ -> MyApp.S3Cloud      # Videos to S3
      "application/pdf" -> MyApp.GCSCloud # Documents to GCS
      _ -> MyApp.S3Cloud                  # Default to S3
    end
  end
end

  
    
  
  Geographic Distribution


Store files in the closest region:
defmodule MyApp.GeoStorage do
  @clouds %{
    "US" => MyApp.USCloud,
    "EU" => MyApp.EUCloud,
    "ASIA" => MyApp.AsiaCloud
  }
  
  def store_for_user(upload, user) do
    cloud = @clouds[user.region] || MyApp.USCloud
    object = Buckets.Object.from_upload(upload)
    cloud.insert(object)
  end
end

  
    
  
  Cross-Cloud Operations



  
    
  
  Copying Between Clouds


defmodule MyApp.CloudSync do
  def copy_object(object, from_cloud, to_cloud) do
    # Load from source
    {:ok, loaded} = from_cloud.load(object)
    
    # Insert to destination
    {:ok, copied} = to_cloud.insert(loaded)
    
    copied
  end
  
  def migrate_bucket(from_cloud, to_cloud) do
    objects = list_all_objects(from_cloud)
    
    objects
    |> Task.async_stream(
      fn object ->
        copy_object(object, from_cloud, to_cloud)
      end,
      max_concurrency: 10,
      timeout: 60_000
    )
    |> Stream.run()
  end
end

  
    
  
  Database Design


Store cloud information with your files:
defmodule MyApp.Files.File do
  use Ecto.Schema
  
  schema "files" do
    field :filename, :string
    field :cloud_module, :string  # "Elixir.MyApp.S3Cloud"
    field :storage_path, :string
    field :content_type, :string
    field :size, :integer
    field :region, :string
    
    timestamps()
  end
  
  def to_object(%__MODULE__{} = file) do
    cloud = String.to_existing_atom(file.cloud_module)
    
    Buckets.Object.new(
      file.id,
      file.filename,
      location: {file.storage_path, cloud},
      metadata: %{
        content_type: file.content_type,
        content_size: file.size
      }
    )
  end
end


  

    
Dynamic Configuration
    

This guide explains how to configure cloud storage dynamically at runtime, enabling multi-tenant applications and flexible storage strategies.

  
    
  
  Overview


Dynamic configuration allows you to:
	Support multi-tenant applications with per-tenant storage
	Switch storage providers without restarting
	Test with different configurations
	Implement feature flags for storage


  
    
  
  Basic Dynamic Configuration



  
    
  
  Process-Scoped Configuration


Set configuration for the current process:
# Set dynamic config
MyApp.Cloud.put_dynamic_config([
  adapter: Buckets.Adapters.S3,
  bucket: "tenant-123-bucket",
  region: "us-west-2",
  access_key_id: "AKIA...",
  secret_access_key: "secret..."
])

# All subsequent operations use this config
{:ok, object} = MyApp.Cloud.insert(upload)

  
    
  
  Scoped Configuration Blocks


Use configuration for a specific block of code:
config = [
  adapter: Buckets.Adapters.GCS,
  bucket: "temporary-bucket",
  service_account_credentials: credentials
]

result = MyApp.Cloud.with_config(config, fn ->
  # All operations in this block use the config
  {:ok, obj1} = MyApp.Cloud.insert(file1)
  {:ok, obj2} = MyApp.Cloud.insert(file2)
  
  [obj1, obj2]
end)

  
    
  
  Multi-Tenant Implementation



  
    
  
  Basic Multi-Tenant Setup


defmodule MyApp.TenantStorage do
  def with_tenant(tenant, fun) do
    config = build_config(tenant)
    MyApp.Cloud.with_config(config, fun)
  end
  
  defp build_config(tenant) do
    case tenant.storage_provider do
      "s3" ->
        [
          adapter: Buckets.Adapters.S3,
          bucket: tenant.s3_bucket,
          region: tenant.s3_region,
          access_key_id: decrypt(tenant.s3_access_key),
          secret_access_key: decrypt(tenant.s3_secret_key)
        ]
        
      "gcs" ->
        [
          adapter: Buckets.Adapters.GCS,
          bucket: tenant.gcs_bucket,
          service_account_credentials: decrypt(tenant.gcs_credentials)
        ]
        
      "volume" ->
        [
          adapter: Buckets.Adapters.Volume,
          bucket: "tenants/#{tenant.id}",
          base_url: tenant.base_url || "http://localhost:4000"
        ]
    end
  end
  
  defp decrypt(encrypted_value) do
    # Decrypt sensitive credentials
    MyApp.Crypto.decrypt(encrypted_value)
  end
end

  
    
  
  Phoenix Integration


Use dynamic config in Phoenix controllers:
defmodule MyAppWeb.FileController do
  plug :load_tenant_config
  
  def upload(conn, %{"file" => upload}) do
    # Config is already set by plug
    object = Buckets.Object.from_upload(upload)
    
    case MyApp.Cloud.insert(object) do
      {:ok, stored} ->
        conn
        |> put_flash(:info, "File uploaded")
        |> redirect(to: ~p"/files")
        
      {:error, reason} ->
        conn
        |> put_flash(:error, "Upload failed: #{inspect(reason)}")
        |> redirect(to: ~p"/files/new")
    end
  end
  
  defp load_tenant_config(conn, _opts) do
    tenant = conn.assigns.current_tenant
    config = MyApp.TenantStorage.build_config(tenant)
    
    MyApp.Cloud.put_dynamic_config(config)
    conn
  end
end

  
    
  
  LiveView Integration


defmodule MyAppWeb.UploadLive do
  use MyAppWeb, :live_view
  
  def mount(_params, session, socket) do
    tenant = get_tenant(session)
    config = MyApp.TenantStorage.build_config(tenant)
    
    # Set config for this LiveView process
    MyApp.Cloud.put_dynamic_config(config)
    
    {:ok,
     socket
     |> assign(:tenant, tenant)
     |> allow_upload(:files, accept: :any, external: &presign_upload/2)}
  end
  
  defp presign_upload(entry, socket) do
    # Uses the dynamic config set in mount
    {:ok, config} = MyApp.Cloud.live_upload(entry)
    {:ok, config, socket}
  end
end

  
    
  
  Background Jobs



  
    
  
  Oban Integration


defmodule MyApp.FileProcessor do
  use Oban.Worker
  
  @impl true
  def perform(%{args: %{"file_id" => file_id, "tenant_id" => tenant_id}}) do
    tenant = Tenants.get!(tenant_id)
    file = Files.get!(file_id)
    
    # Run with tenant's config
    MyApp.TenantStorage.with_tenant(tenant, fn ->
      # Load file data
      {:ok, object} = MyApp.Cloud.load(file.object)
      
      # Process file
      process_file(object)
      
      # Save results
      {:ok, processed} = MyApp.Cloud.insert(processed_object)
      
      Files.update(file, %{processed_path: processed.location.path})
    end)
    
    :ok
  end
end


  

    
Error Handling
    

This guide covers error handling strategies for robust file storage operations with Buckets.

  
    
  
  Error Types



  
    
  
  Common Errors


Buckets operations can return various error types:
# Configuration errors
{:error, [:bucket, :access_key_id]}  # Missing required fields

# Network errors
{:error, %Mint.TransportError{}}     # Connection failed
{:error, {:http_error, 500}}         # Server error

# Permission errors
{:error, :unauthorized}              # Invalid credentials
{:error, :forbidden}                 # Insufficient permissions

# Resource errors
{:error, :not_found}                 # Object doesn't exist
{:error, :already_exists}            # Duplicate object

# Validation errors
{:error, :file_too_large}            # Size limit exceeded
{:error, :invalid_content_type}      # Type not allowed

  
    
  
  Basic Error Handling



  
    
  
  Pattern Matching


case MyApp.Cloud.insert(object) do
  {:ok, stored} ->
    # Success path
    {:ok, stored}
    
  {:error, :unauthorized} ->
    # Handle auth error
    Logger.error("Invalid cloud storage credentials")
    {:error, "Authentication failed"}
    
  {:error, %{status: 507}} ->
    # Storage full
    Logger.error("Cloud storage quota exceeded")
    {:error, "Storage limit reached"}
    
  {:error, reason} ->
    # Generic error
    Logger.error("Upload failed: #{inspect(reason)}")
    {:error, "Upload failed"}
end

  
    
  
  Using Bang Functions


def upload_or_raise(file_path) do
  try do
    object = Buckets.Object.from_file(file_path)
    stored = MyApp.Cloud.insert!(object)
    {:ok, stored}
  rescue
    e in RuntimeError ->
      Logger.error("Upload failed: #{e.message}")
      reraise e, __STACKTRACE__
  end
end

  
    
  
  Retry Strategies



  
    
  
  Simple Retry


defmodule MyApp.Storage.Retry do
  def with_retry(fun, opts \\ []) do
    max_attempts = Keyword.get(opts, :max_attempts, 3)
    delay = Keyword.get(opts, :delay, 1000)
    
    do_retry(fun, max_attempts, delay, 1)
  end
  
  defp do_retry(fun, max_attempts, delay, attempt) do
    case fun.() do
      {:ok, result} ->
        {:ok, result}
        
      {:error, reason} when attempt < max_attempts ->
        if retryable_error?(reason) do
          Process.sleep(delay * attempt)  # Exponential backoff
          do_retry(fun, max_attempts, delay, attempt + 1)
        else
          {:error, reason}
        end
        
      {:error, reason} ->
        {:error, reason}
    end
  end
  
  defp retryable_error?(:timeout), do: true
  defp retryable_error?({:http_error, status}) when status >= 500, do: true
  defp retryable_error?(%Mint.TransportError{}), do: true
  defp retryable_error?(_), do: false
end

# Usage
MyApp.Storage.Retry.with_retry(fn ->
  MyApp.Cloud.insert(object)
end, max_attempts: 5, delay: 2000)

  
    
  
  User-Friendly Errors



  
    
  
  Error Translation


defmodule MyApp.Storage.ErrorMessages do
  def translate({:error, :unauthorized}) do
    "Storage authentication failed. Please check your credentials."
  end
  
  def translate({:error, :not_found}) do
    "The requested file could not be found."
  end
  
  def translate({:error, {:http_error, 507}}) do
    "Storage quota exceeded. Please upgrade your plan."
  end
  
  def translate({:error, :timeout}) do
    "The operation timed out. Please try again."
  end
  
  def translate({:error, :file_too_large}) do
    "The file is too large. Maximum size is 10MB."
  end
  
  def translate(_error) do
    "An unexpected error occurred. Please try again later."
  end
end

# In controller
case MyApp.Cloud.insert(object) do
  {:ok, _} ->
    put_flash(conn, :info, "Upload successful")
    
  {:error, _} = error ->
    message = MyApp.Storage.ErrorMessages.translate(error)
    put_flash(conn, :error, message)
end


  

    
Testing
    

This guide covers testing strategies for applications using Buckets, including unit tests, integration tests, and testing with live cloud services.

  
    
  
  Test Setup



  
    
  
  Basic Test Configuration


# config/test.exs
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.Volume,
  bucket: "tmp/test_storage",
  base_url: "http://localhost:4001"

# Ensure test isolation
config :my_app, :storage_base_path, "tmp/test_#{System.get_pid()}"

  
    
  
  Test Helper Module


defmodule MyApp.StorageCase do
  use ExUnit.CaseTemplate
  
  using do
    quote do
      import MyApp.StorageCase
      
      setup :create_test_storage
    end
  end
  
  def create_test_storage(_context) do
    # Create unique test directory
    test_dir = "tmp/test_#{:rand.uniform(1_000_000)}"
    File.mkdir_p!(test_dir)
    
    # Configure for this test
    config = [
      adapter: Buckets.Adapters.Volume,
      bucket: test_dir,
      base_url: "http://localhost:4001"
    ]
    
    MyApp.Cloud.put_dynamic_config(config)
    
    on_exit(fn ->
      # Cleanup
      File.rm_rf!(test_dir)
    end)
    
    {:ok, storage_dir: test_dir}
  end
  
  def create_test_object(attrs \\ %{}) do
    defaults = %{
      uuid: Ecto.UUID.generate(),
      filename: "test.pdf",
      content: "test content",
      content_type: "application/pdf"
    }
    
    attrs = Map.merge(defaults, attrs)
    
    Buckets.Object.new(
      attrs.uuid,
      attrs.filename,
      metadata: %{
        content_type: attrs.content_type,
        content_size: byte_size(attrs.content)
      }
    )
  end
end

  
    
  
  Unit Tests



  
    
  
  Testing Storage Operations


defmodule MyApp.CloudTest do
  use MyApp.StorageCase
  
  describe "insert/1" do
    test "stores object successfully", %{storage_dir: dir} do
      object = create_test_object()
      
      assert {:ok, stored} = MyApp.Cloud.insert(object)
      assert stored.stored?
      assert stored.location.path =~ object.uuid
      
      # Verify file exists
      file_path = Path.join([dir, stored.location.path])
      assert File.exists?(file_path)
    end
    
    test "handles missing file" do
      object = Buckets.Object.from_file("non_existent.pdf")
      
      assert_raise RuntimeError, ~r/non-existent file/, fn ->
        MyApp.Cloud.insert!(object)
      end
    end
  end
  
  describe "read/1" do
    test "reads stored object data" do
      object = create_test_object(content: "Hello, World!")
      {:ok, stored} = MyApp.Cloud.insert(object)
      
      assert {:ok, data} = MyApp.Cloud.read(stored)
      assert data == "Hello, World!"
    end
    
    test "returns error for missing object" do
      fake_object = create_test_object()
      |> Map.put(:location, Buckets.Location.new("fake/path", MyApp.Cloud))
      |> Map.put(:stored?, true)
      
      assert {:error, :not_found} = MyApp.Cloud.read(fake_object)
    end
  end
end

  
    
  
  LiveView Testing



  
    
  
  Testing Direct Uploads


defmodule MyAppWeb.UploadLiveTest do
  use MyAppWeb.ConnCase
  use MyApp.StorageCase
  import Phoenix.LiveViewTest
  
  describe "file uploads" do
    test "direct upload to cloud storage", %{conn: conn} do
      {:ok, view, _html} = live(conn, ~p"/uploads/new")
      
      # Simulate file selection
      file =
        file_input(view, "form", :document, [
          %{
            name: "invoice.pdf",
            content: File.read!("test/fixtures/sample.pdf"),
            type: "application/pdf"
          }
        ])
      
      # Trigger upload
      assert render_upload(file, "invoice.pdf")
      
      # Submit form
      html = form(view, "form") |> render_submit()
      
      assert html =~ "Upload successful"
      assert MyApp.Files.count() == 1
    end
  end
end

  
    
  
  Mocking and Stubbing



  
    
  
  Using Mox for Cloud Operations


# test/support/mocks.ex
Mox.defmock(MyApp.CloudMock, for: Buckets.Cloud)

# In tests
defmodule MyApp.ServiceTest do
  use ExUnit.Case
  import Mox
  
  setup :verify_on_exit!
  
  test "processes file after upload" do
    object = create_test_object()
    
    expect(MyApp.CloudMock, :insert, fn ^object ->
      {:ok, %{object | stored?: true}}
    end)
    
    expect(MyApp.CloudMock, :read, fn _object ->
      {:ok, "processed content"}
    end)
    
    assert {:ok, result} = MyApp.FileProcessor.process(object)
    assert result.content == "processed content"
  end
end

  
    
  
  Running Tests


# Run all tests
mix test

# Run only unit tests (exclude external)
mix test --exclude external

# Run specific adapter tests
mix test --only s3

# Run with coverage
mix test --cover



  

    
Telemetry and Monitoring
    

This guide covers monitoring and observability for Buckets operations using Telemetry.

  
    
  
  Overview


Buckets emits telemetry events for all major operations, allowing you to monitor performance, track errors, and gather metrics about your file storage operations.

  
    
  
  Available Events



  
    
  
  Cloud Operations


	[:buckets, :cloud, :insert, :start] - File upload started
	[:buckets, :cloud, :insert, :stop] - File upload completed
	[:buckets, :cloud, :insert, :exception] - File upload failed
	[:buckets, :cloud, :read, :start] - File read started
	[:buckets, :cloud, :read, :stop] - File read completed
	[:buckets, :cloud, :delete, :start] - File deletion started
	[:buckets, :cloud, :delete, :stop] - File deletion completed
	[:buckets, :cloud, :url, :start] - URL generation started
	[:buckets, :cloud, :url, :stop] - URL generation completed


  
    
  
  Adapter Operations


	[:buckets, :adapter, :put, :start] - Adapter upload started
	[:buckets, :adapter, :put, :stop] - Adapter upload completed
	[:buckets, :adapter, :get, :start] - Adapter download started
	[:buckets, :adapter, :get, :stop] - Adapter download completed


  
    
  
  Basic Setup


defmodule MyApp.Telemetry do
  require Logger

  def setup do
    events = [
      [:buckets, :cloud, :insert, :start],
      [:buckets, :cloud, :insert, :stop],
      [:buckets, :cloud, :insert, :exception],
      [:buckets, :cloud, :read, :stop],
      [:buckets, :cloud, :delete, :stop],
      [:buckets, :cloud, :url, :stop]
    ]

    :telemetry.attach_many(
      "buckets-handler",
      events,
      &handle_event/4,
      nil
    )
  end

  defp handle_event([:buckets, :cloud, :insert, :stop], measurements, metadata, _config) do
    Logger.info("File uploaded",
      duration_ms: measurements.duration / 1_000_000,
      filename: metadata.filename,
      cloud: metadata.cloud_module
    )
  end

  defp handle_event([:buckets, :cloud, :insert, :exception], _measurements, metadata, _config) do
    Logger.error("File upload failed",
      error: inspect(metadata.error),
      filename: metadata.filename
    )
  end

  # Handle other events...
end

  
    
  
  Metrics Collection


For detailed metrics and monitoring, integrate with your preferred metrics system.

  
    
  
  StatsD Integration


defmodule MyApp.Metrics.StatsD do
  def setup do
    :telemetry.attach_many(
      "buckets-statsd",
      buckets_events(),
      &handle_event/4,
      nil
    )
  end

  defp handle_event([:buckets, :cloud, event, :stop], measurements, metadata, _config) do
    # Record timing
    StatsD.histogram(
      "buckets.cloud.#{event}.duration",
      measurements.duration / 1_000_000,
      tags: ["adapter:#{metadata.adapter}", "cloud:#{cloud_name(metadata)}"]
    )

    # Count operations
    StatsD.increment(
      "buckets.cloud.#{event}.count",
      tags: ["adapter:#{metadata.adapter}"]
    )
  end

  defp handle_event([:buckets, :cloud, event, :exception], _measurements, metadata, _config) do
    StatsD.increment(
      "buckets.cloud.#{event}.error",
      tags: ["adapter:#{metadata.adapter}", "error:#{error_type(metadata.error)}"]
    )
  end
end

  
    
  
  Prometheus Integration


defmodule MyApp.Metrics.Prometheus do
  use Prometheus.Metric

  def setup do
    # Define metrics
    Histogram.new(
      name: :buckets_operation_duration_seconds,
      help: "Duration of bucket operations",
      labels: [:operation, :adapter, :cloud],
      buckets: [0.01, 0.05, 0.1, 0.5, 1, 5, 10]
    )

    Counter.new(
      name: :buckets_operations_total,
      help: "Total number of bucket operations",
      labels: [:operation, :adapter, :cloud, :status]
    )

    # Attach handlers
    :telemetry.attach_many(
      "buckets-prometheus",
      buckets_events(),
      &handle_event/4,
      nil
    )
  end

  defp handle_event([:buckets, :cloud, operation, :stop], measurements, metadata, _config) do
    Histogram.observe(
      [name: :buckets_operation_duration_seconds],
      measurements.duration / 1_000_000_000,
      labels: [operation, metadata.adapter, cloud_name(metadata)]
    )

    Counter.inc(
      [name: :buckets_operations_total],
      labels: [operation, metadata.adapter, cloud_name(metadata), "success"]
    )
  end
end

  
    
  
  Custom Metrics


Add your own metrics around Buckets operations:
defmodule MyApp.Storage.Metrics do
  def upload_with_metrics(file_path) do
    start_time = System.monotonic_time()
    file_size = File.stat!(file_path).size

    metadata = %{
      file_path: file_path,
      file_size: file_size
    }

    :telemetry.span(
      [:my_app, :storage, :upload],
      metadata,
      fn ->
        object = Buckets.Object.from_file(file_path)
        result = MyApp.Cloud.insert(object)
        
        {result, Map.put(metadata, :stored?, match?({:ok, _}, result))}
      end
    )
  end
end

  
    
  
  Performance Monitoring


Track performance across different adapters and operations:
defmodule MyApp.PerformanceMonitor do
  use GenServer

  def start_link(_opts) do
    GenServer.start_link(__MODULE__, %{}, name: __MODULE__)
  end

  def init(_state) do
    :telemetry.attach_many(
      "performance-monitor",
      [
        [:buckets, :adapter, :put, :stop],
        [:buckets, :adapter, :get, :stop]
      ],
      &handle_event/4,
      nil
    )

    {:ok, %{}}
  end

  defp handle_event([_, _, operation, :stop], measurements, metadata, _config) do
    GenServer.cast(__MODULE__, {:record, operation, measurements, metadata})
  end

  def handle_cast({:record, operation, measurements, metadata}, state) do
    key = {metadata.adapter, operation}
    duration_ms = measurements.duration / 1_000_000

    # Alert if slow
    if duration_ms > 5000 do
      Logger.warn("Slow #{operation} operation",
        adapter: metadata.adapter,
        duration_ms: duration_ms,
        path: metadata.path
      )
    end

    # Update rolling average
    new_state = Map.update(state, key, [duration_ms], fn durations ->
      [duration_ms | Enum.take(durations, 99)]
    end)

    {:noreply, new_state}
  end
end

  
    
  
  Error Tracking


Monitor and alert on errors:
defmodule MyApp.ErrorTracker do
  def setup do
    :telemetry.attach(
      "error-tracker",
      [:buckets, :cloud, :insert, :exception],
      &track_error/4,
      nil
    )
  end

  defp track_error(_event, _measurements, metadata, _config) do
    Sentry.capture_exception(metadata.error,
      stacktrace: metadata.stacktrace,
      extra: %{
        filename: metadata.filename,
        cloud_module: metadata.cloud_module,
        adapter: metadata.adapter
      }
    )
  end
end

  
    
  
  Dashboards


Create dashboards to visualize your storage metrics:

  
    
  
  Grafana Dashboard


Key panels to include:
	Upload/download rate
	Operation duration percentiles
	Error rate by adapter
	Storage usage trends
	Concurrent operations


  
    
  
  Example Query (Prometheus)


# Upload success rate
rate(buckets_operations_total{operation="insert",status="success"}[5m])
/
rate(buckets_operations_total{operation="insert"}[5m])

# P95 upload duration
histogram_quantile(0.95,
  rate(buckets_operation_duration_seconds_bucket{operation="insert"}[5m])
)

  
    
  
  Best Practices


	Sample high-volume events - Avoid overwhelming metrics systems
	Use consistent labels - Maintain standard label naming
	Alert on anomalies - Set up alerts for error spikes
	Track business metrics - Not just technical metrics
	Monitor costs - Track usage that affects billing
	Regular review - Analyze metrics for optimization opportunities


  
    
  
  Next Steps


	Implement Security Monitoring
	Set up Performance Optimization
	Configure Production Monitoring



  

    
Buckets 
    



      
Cloud-agnostic file storage for Elixir with Phoenix integration.
Buckets provides a unified API for working with file storage across different
cloud providers. Whether you're using local filesystem storage for development,
Google Cloud Storage, Amazon S3, or other S3-compatible providers, Buckets
offers a consistent interface with powerful features.

  
    
  
  Features


	Multiple Storage Adapters - Support for filesystem, Google Cloud Storage,
Amazon S3, and S3-compatible providers (Cloudflare R2, DigitalOcean Spaces, Tigris)
	Direct Uploads - Upload files directly from browsers to cloud storage,
bypassing your Phoenix server
	Signed URLs - Generate time-limited, secure URLs for private files
	LiveView Integration - Seamless integration with Phoenix LiveView's
file upload functionality
	Dynamic Configuration - Switch storage providers at runtime for
multi-tenant applications
	Development Tools - Built-in router for local file uploads/downloads
during development
	Telemetry - Comprehensive instrumentation for monitoring and debugging


  
    
  
  Quick Start


	Add buckets to your dependencies:
 def deps do
   [{:buckets, "~> 1.0.0-rc.4"}]
 end

	Create a Cloud module:
 defmodule MyApp.Cloud do
   use Buckets.Cloud, otp_app: :my_app
 end

	Configure your adapter:
 # config/dev.exs
 config :my_app, MyApp.Cloud,
   adapter: Buckets.Adapters.Volume,
   bucket: "priv/uploads",
   base_url: "http://localhost:4000"

	Upload files:
 # From a Plug.Upload or Phoenix.LiveView.UploadEntry
 {:ok, object} = MyApp.Cloud.insert(upload)

 # From a file path
 object = Buckets.Object.from_file("photo.jpg")
 {:ok, stored} = MyApp.Cloud.insert(object)

 # Generate a signed URL
 {:ok, url} = MyApp.Cloud.url(stored, expires: 3600)



  
    
  
  Storage Adapters


Buckets includes these built-in adapters:
	Buckets.Adapters.Volume - Local filesystem storage
	Buckets.Adapters.S3 - Amazon S3 and S3-compatible services
	Buckets.Adapters.GCS - Google Cloud Storage

See the adapter modules for specific configuration options.

  
    
  
  Next Steps


	See Buckets.Cloud for the high-level API
	See Buckets.Object for working with file objects
	Read the Getting Started guide
	Learn about Direct Uploads with LiveView

Buckets.Cloud behaviour
    



      
Defines a cloud.
A cloud manages the movement of files and data between your application and
a remote bucket for persistent storage.
When used, it expects :otp_app as an option:
defmodule MyApp.Cloud do
  use Buckets.Cloud, otp_app: :my_app
end
Configuration is fetched from the application config, using a combination of
:otp_app and the module that you defined:
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.Volume,
  bucket: "tmp/buckets_volume",
  base_url: "http://localhost:4000"
Each Cloud module corresponds to a single storage backend, similar to how
Ecto.Repo modules correspond to a single database. For multi-cloud applications,
define multiple Cloud modules:
defmodule MyApp.VolumeCloud do
  use Buckets.Cloud, otp_app: :my_app
end

defmodule MyApp.GCSCloud do
  use Buckets.Cloud, otp_app: :my_app
end

defmodule MyApp.S3Cloud do
  use Buckets.Cloud, otp_app: :my_app
end
You may also specify config dynamically at runtime, using the :config opt
where it is supported.

  
    
  
  Supervision


Cloud modules start a supervisor that manages any background processes required
by the configured adapter. Some adapters (like GCS) need authentication servers,
while others (like Volume, S3) don't need any supervised processes.
Only add Cloud modules to your supervision tree if they need background processes.
If you add a Cloud module that doesn't need supervision (Volume, S3), you'll see
a warning message suggesting you remove it to avoid unnecessary overhead.
# Only needed for adapters that require background processes (like GCS)
children = [
  MyApp.GCSCloud  # GCS needs auth servers
  # MyApp.VolumeCloud - Not needed, would show warning
  # MyApp.S3Cloud - Not needed, would show warning
]

  
    
  
  Dynamic Configuration


For multi-tenant applications where cloud configurations are determined at runtime,
every Cloud module supports dynamic configuration using the process dictionary,
similar to Ecto's dynamic repositories.

  
    
  
  Usage


There are two ways to use dynamic configuration:
1. Scoped Configuration (like Ecto transactions)
Use with_config/2 for temporary configuration:
# Define the runtime configuration
config = [
  adapter: Buckets.Adapters.S3,
  bucket: "user-bucket",
  access_key_id: "AKIA...",
  secret_access_key: "secret...",
  region: "us-east-1"
]

# Execute operations with the dynamic config
{:ok, object} = MyApp.Cloud.with_config(config, fn ->
  MyApp.Cloud.insert("file.pdf")
  MyApp.Cloud.insert("another.pdf")  # Same config
end)
2. Process-Local Configuration (like Ecto.Repo.put_dynamic_repo)
Use put_dynamic_config/1 for persistent configuration in the current process:
# Set dynamic config for this process
:ok = MyApp.Cloud.put_dynamic_config([
  adapter: Buckets.Adapters.GCS,
  bucket: "tenant-specific-bucket",
  service_account_credentials: tenant.credentials
])

# All subsequent operations use the dynamic config
{:ok, object1} = MyApp.Cloud.insert("file1.pdf")
{:ok, object2} = MyApp.Cloud.insert("file2.pdf")

  
    
  
  Auth Server Management


Auth servers (for GCS) are automatically started and cached per-process as needed.
You don't need any special configuration or supervision setup for dynamic clouds.

Buckets.Location 
    



      
Represents a storage location for a Buckets.Object.
The config field can be either:
	A keyword list with adapter configuration
	A Cloud module atom that implements the Buckets.Cloud behaviour

Buckets.Object 
    



      
Represents a file object in the Buckets storage system.
An Object encapsulates file data, metadata, and location information. Objects
can exist in various states:

  
    
  
  Object States


	Local only - Created from a file or upload, not yet stored remotely
	stored?: false
	location: %Buckets.Location.NotConfigured{}
	data: {:file, path} or {:data, binary}


	Stored remotely - Uploaded to cloud storage
	stored?: true
	location: %Buckets.Location{}
	data: nil (unless loaded)


	Loaded - Remote object with data fetched
	stored?: true
	location: %Buckets.Location{}
	data: {:file, path} or {:data, binary}




  
    
  
  Creating Objects


Objects can be created from:
	Files on disk: from_file/1
	Plug uploads: from_upload/1 with %Plug.Upload{}
	LiveView uploads: from_upload/1 with %Phoenix.LiveView.UploadEntry{}
	Manually: new/3


  
    
  
  Metadata


Objects store metadata including:
	:content_type - MIME type of the file
	:content_size - Size in bytes
	Custom metadata can be added as needed


  
    
  
  Example Usage


# Create from file
object = Buckets.Object.from_file("/path/to/document.pdf")

# Upload to storage
{:ok, stored} = MyApp.Cloud.insert(object)

# Read data from stored object
{:ok, data} = MyApp.Cloud.read(stored)

# Load remote object data
{:ok, loaded} = MyApp.Cloud.load(stored)
{:ok, data} = Buckets.Object.read(loaded)

Buckets.SignedURL 
    



      
Represents a signed URL for direct access to objects in cloud storage.
Signed URLs provide temporary, secure access to objects without requiring
authentication. They are commonly used for:
	Direct file downloads from cloud storage
	Direct uploads to cloud storage (especially with LiveView)
	Temporary sharing of private files


  
    
  
  Structure


A SignedURL contains:
	:url - The complete signed URL string that can be used for access
	:location - The Buckets.Location struct indicating where the object is stored


  
    
  
  Usage


SignedURLs are typically generated by adapters through the url/2 callback:
{:ok, signed_url} = MyApp.Cloud.url(object, expires_in: 3600)

# Access the URL string
url_string = signed_url.url
# or use String.Chars protocol
url_string = to_string(signed_url)

  
    
  
  LiveView Uploads


SignedURLs are used for direct uploads in LiveView:
{:ok, config} = MyApp.Cloud.live_upload(entry)
# config.url is a SignedURL for uploading

  
    
  
  JSON Encoding


SignedURLs implement the Jason.Encoder protocol and encode to just the URL string:
Jason.encode!(%{upload_url: signed_url})
# {"upload_url": "https://storage.example.com/bucket/file?signature=..."}

  
    
  
  Security Considerations


	Signed URLs typically expire after a set time period
	They should be treated as sensitive data
	Different adapters have different expiration limits and security features

Buckets.Adapters.GCS 
    



      
Google Cloud Storage adapter for Buckets.
This adapter provides a native implementation for GCS operations using
only :req as a dependency.

  
    
  
  Setup


To use this adapter, you need to start your Cloud module in your application's
supervision tree to enable automatic token caching and refresh:
children = [
  # ... your other processes
  MyApp.Cloud
]

Supervisor.start_link(children, opts)
The Cloud module will automatically start the required authentication processes for GCS locations.
The supervisor will automatically manage authentication tokens for each unique set of
service account credentials, refreshing them before they expire.

  
    
  
  Configuration


You can configure GCS using either service account credentials as a JSON string
or by providing a path to a service account JSON file:
# Using credentials from environment variable (JSON string)
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.GCS,
  bucket: "my-bucket",
  service_account_credentials: System.fetch_env!("GOOGLE_CREDENTIALS")

# Using credentials from file path
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.GCS,
  bucket: "my-bucket",
  service_account_path: "/path/to/service-account.json"

  
    
  
  Optional Configuration


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.GCS,
  bucket: "my-bucket",
  service_account_credentials: System.fetch_env!("GOOGLE_CREDENTIALS"),
  # Optional: base path within bucket
  path: "uploads",
  # Optional: uploader module for LiveView direct uploads
  uploader: "GCS"
The :service_account_credentials option accepts a JSON string, making it easy to pass
credentials via environment variables in production environments.

  
    
  
  Performance


This implementation automatically caches and refreshes Google Cloud access tokens,
significantly reducing latency compared to generating new tokens for each request.
Tokens are refreshed 5 minutes before expiration to ensure uninterrupted service.

Buckets.Adapters.S3 
    



      
Adapter for Amazon S3 and S3-compatible storage services.
This adapter supports multiple S3-compatible providers through the :provider option:
	:aws (default) - Amazon S3
	:cloudflare - Cloudflare R2
	:digitalocean - DigitalOcean Spaces
	:tigris - Tigris


  
    
  
  Configuration



  
    
  
  Amazon S3 (default)


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  bucket: "my-bucket",
  region: "us-east-1",
  access_key_id: System.fetch_env!("AWS_ACCESS_KEY_ID"),
  secret_access_key: System.fetch_env!("AWS_SECRET_ACCESS_KEY")

  
    
  
  Cloudflare R2


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  provider: :cloudflare,
  bucket: "my-bucket",
  endpoint_url: "https://YOUR_ACCOUNT_ID.r2.cloudflarestorage.com",
  access_key_id: System.fetch_env!("R2_ACCESS_KEY_ID"),
  secret_access_key: System.fetch_env!("R2_SECRET_ACCESS_KEY")
Note: Cloudflare R2 automatically uses region: "auto".

  
    
  
  DigitalOcean Spaces


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  provider: :digitalocean,
  bucket: "my-bucket",
  access_key_id: System.fetch_env!("SPACES_ACCESS_KEY"),
  secret_access_key: System.fetch_env!("SPACES_SECRET_KEY")
Note: DigitalOcean Spaces defaults to NYC3 region and endpoint. Override with:
config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  provider: :digitalocean,
  bucket: "my-bucket",
  region: "sfo3",
  endpoint_url: "https://sfo3.digitaloceanspaces.com",
  access_key_id: System.fetch_env!("SPACES_ACCESS_KEY"),
  secret_access_key: System.fetch_env!("SPACES_SECRET_KEY")

  
    
  
  Tigris


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.S3,
  provider: :tigris,
  bucket: "my-bucket",
  region: "auto",
  access_key_id: System.fetch_env!("TIGRIS_ACCESS_KEY_ID"),
  secret_access_key: System.fetch_env!("TIGRIS_SECRET_ACCESS_KEY")
Note: Tigris automatically uses the endpoint https://fly.storage.tigris.dev.

  
    
  
  Required Options


	:bucket - The S3 bucket name
	:access_key_id - AWS access key ID or equivalent
	:secret_access_key - AWS secret access key or equivalent
	:region - AWS region (defaults to "auto" for some providers)
	:endpoint_url - Required for non-AWS providers


  
    
  
  Optional Options


	:provider - The S3-compatible provider (defaults to :aws)
	:path - Base path within the bucket
	:uploader - Uploader configuration for direct uploads


  
    
  
  Supervision


This adapter does not require any supervised processes. You do not need to add Cloud modules
using this adapter to your supervision tree.

Buckets.Adapters.Volume 
    



      
Adapter for local filesystem storage.
The Volume adapter stores files on the local filesystem and is primarily intended
for development environments. It supports signed URLs and integrates with the
Buckets dev router for handling uploads and downloads.

  
    
  
  Configuration


config :my_app, MyApp.Cloud,
  adapter: Buckets.Adapters.Volume,
  bucket: "tmp/buckets_volume",
  base_url: "http://localhost:4000"

  
    
  
  Required Options


	:bucket - The local directory path where files will be stored
	:base_url - The base URL for generating file URLs (e.g., your Phoenix app URL)


  
    
  
  Optional Options


	:path - Base path within the bucket directory
	:endpoint - Custom endpoint path for signed URLs (defaults to "/buckets")
	:uploader - Uploader configuration for direct uploads


  
    
  
  Development Router Integration


To handle local uploads and downloads in development, add the Volume routes to your router:
# In your router.ex
if Application.compile_env(:my_app, :dev_routes) do
  import Buckets.Router

  buckets_volume(MyApp.Cloud)
end
This will mount routes at /__buckets__/:bucket/*path for handling:
	File uploads (PUT requests)
	File downloads (GET requests)


  
    
  
  Signed URLs


The Volume adapter supports signed URLs for secure file uploads and downloads:
{:ok, object} = MyApp.Cloud.url(object, expires_in: 3600)
# Returns object with signed URL valid for 1 hour

  
    
  
  Supervision


This adapter does not require any supervised processes. Do not add Cloud modules
using this adapter to your supervision tree.

Buckets.Adapter behaviour
    



      
Behaviour definition for Buckets storage adapters.
An adapter implements the storage operations for a specific backend (filesystem,
S3, GCS, etc.). All adapters must implement the required callbacks to handle
file operations like uploading, downloading, and generating signed URLs.

  
    
  
  Implementing an Adapter


To create a new adapter, implement this behaviour:
defmodule MyApp.CustomAdapter do
  @behaviour Buckets.Adapter

  @impl true
  def validate_config(config) do
    # Validate and normalize configuration
  end

  @impl true
  def put(object, remote_path, config) do
    # Upload object to storage
  end

  @impl true
  def get(remote_path, config) do
    # Download data from storage
  end

  @impl true
  def url(remote_path, config) do
    # Generate signed URL for direct access
  end

  @impl true
  def delete(remote_path, config) do
    # Delete object from storage
  end

  # Optional: Only if adapter needs supervised processes
  @impl true
  def child_spec(config, cloud_module) do
    # Return supervisor child specification
  end
end

  
    
  
  Configuration


Each adapter defines its own configuration options, but common ones include:
	:bucket - The storage bucket/container name
	:path - Base path within the bucket
	:uploader - Uploader type for LiveView direct uploads


  
    
  
  Supervised Processes


Some adapters (like GCS) require background processes for authentication or
connection management. These adapters should implement the optional child_spec/2
callback. Adapters that don't need supervision (like Volume, S3) should not
implement this callback.

Buckets.Router 
    



      
Router helpers for integrating Buckets with Phoenix applications.
This module provides macros for setting up development routes that handle
file uploads and downloads when using the Volume adapter. These routes
enable direct file uploads to the local filesystem during development,
mimicking the behavior of cloud storage providers.

  
    
  
  Usage


Import this module in your Phoenix router and use the buckets_volume/2
macro to set up the necessary routes:
# In your router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  # Only include in development
  if Application.compile_env(:my_app, :dev_routes) do
    import Buckets.Router

    # Sets up routes at /__buckets__/:bucket/*path
    buckets_volume(MyApp.Cloud)
  end
end

  
    
  
  Routes Created


The macro creates two routes under the /__buckets__ scope:
	GET /__buckets__/:bucket/*path - Download files
	PUT /__buckets__/:bucket/*path - Upload files


  
    
  
  Security


These routes validate:
	The bucket name matches the configured bucket
	Upload requests have valid signatures (generated by the Volume adapter)


  
    
  
  Configuration


The routes can be customized with options:
buckets_volume(MyApp.Cloud, cache_control: 3600)
Available options are passed to the controller and may include:
	:cache_control - Max age in seconds for cache control headers

Buckets.Router.VolumeController 
    



      
Phoenix controller for handling local file uploads and downloads in development.
This controller is automatically used by the Buckets.Router.buckets_volume/2
macro to handle Volume adapter file operations.

  
    
  
  Endpoints


	get/2 - Downloads files from the local filesystem
	put/2 - Uploads files to the local filesystem with signature validation

For usage and configuration, see Buckets.Router.

Buckets.Telemetry 
    



      
Telemetry events for Buckets.
This module defines telemetry events that other applications can attach to.

  
    
  
  Event Format


Events follow the convention: [:buckets, component, operation, stage]

  
    
  
  Events



  
    
  
  Cloud Operations


	[:buckets, :cloud, :insert, :start] - Emitted when a file upload operation starts
	[:buckets, :cloud, :insert, :stop] - Emitted when a file upload operation completes
	[:buckets, :cloud, :insert, :exception] - Emitted when a file upload operation raises an exception
	[:buckets, :cloud, :delete, :start] - Emitted when a file deletion operation starts
	[:buckets, :cloud, :delete, :stop] - Emitted when a file deletion operation completes
	[:buckets, :cloud, :delete, :exception] - Emitted when a file deletion operation raises an exception
	[:buckets, :cloud, :read, :start] - Emitted when a file read operation starts
	[:buckets, :cloud, :read, :stop] - Emitted when a file read operation completes
	[:buckets, :cloud, :read, :exception] - Emitted when a file read operation raises an exception
	[:buckets, :cloud, :load, :start] - Emitted when a file load operation starts
	[:buckets, :cloud, :load, :stop] - Emitted when a file load operation completes
	[:buckets, :cloud, :load, :exception] - Emitted when a file load operation raises an exception
	[:buckets, :cloud, :url, :start] - Emitted when a URL generation operation starts
	[:buckets, :cloud, :url, :stop] - Emitted when a URL generation operation completes
	[:buckets, :cloud, :url, :exception] - Emitted when a URL generation operation raises an exception


  
    
  
  Adapter Operations


	[:buckets, :adapter, :put, :start] - Emitted when an adapter put operation starts
	[:buckets, :adapter, :put, :stop] - Emitted when an adapter put operation completes
	[:buckets, :adapter, :put, :exception] - Emitted when an adapter put operation raises an exception
	[:buckets, :adapter, :get, :start] - Emitted when an adapter get operation starts
	[:buckets, :adapter, :get, :stop] - Emitted when an adapter get operation completes
	[:buckets, :adapter, :get, :exception] - Emitted when an adapter get operation raises an exception
	[:buckets, :adapter, :url, :start] - Emitted when an adapter URL generation starts
	[:buckets, :adapter, :url, :stop] - Emitted when an adapter URL generation completes
	[:buckets, :adapter, :url, :exception] - Emitted when an adapter URL generation raises an exception
	[:buckets, :adapter, :delete, :start] - Emitted when an adapter delete operation starts
	[:buckets, :adapter, :delete, :stop] - Emitted when an adapter delete operation completes
	[:buckets, :adapter, :delete, :exception] - Emitted when an adapter delete operation raises an exception


  
    
  
  Authentication Operations


	[:buckets, :auth, :token, :fetch, :start] - Emitted when token fetching starts
	[:buckets, :auth, :token, :fetch, :stop] - Emitted when token fetching completes
	[:buckets, :auth, :token, :fetch, :exception] - Emitted when token fetching raises an exception
	[:buckets, :auth, :token, :refresh, :start] - Emitted when token refresh starts
	[:buckets, :auth, :token, :refresh, :stop] - Emitted when token refresh completes
	[:buckets, :auth, :token, :refresh, :exception] - Emitted when token refresh raises an exception


  
    
  
  Usage


To attach to these events in your application:
:telemetry.attach(
  "my-handler-id",
  [:buckets, :cloud, :insert, :start],
  &MyApp.handle_cloud_insert/4,
  nil
)

def handle_cloud_insert(_event_name, measurements, metadata, _config) do
  # Process the event
  IO.inspect(measurements)
  IO.inspect(metadata)
end
You may attach a default logger handler, if you want basic logs for all telemetry events:
Buckets.Telemetry.attach_default_logger(:debug)

Buckets.Util 
    



      
Utility functions for the Buckets library.
This module provides helper functions used internally by Buckets for
file operations and string manipulation.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Buckets.Adapters.GCS.Auth - Buckets v1.0.0-rc.4
    
    

    


  
  

    
Buckets.Adapters.GCS.Auth 
    



      
Google Cloud Storage authentication using service account credentials.
Handles JWT token generation and OAuth2 token exchange without external dependencies.

Buckets.Adapters.GCS.AuthServer 
    



      
GenServer that manages Google Cloud Storage access tokens.
Automatically refreshes tokens before they expire to ensure uninterrupted access to GCS APIs.

Buckets.Adapters.GCS.Signature 
    



      
Google Cloud Storage V4 signed URL generation.
Implements the V4 signing process for creating signed URLs that allow
temporary access to GCS objects without requiring authentication.

Buckets.Cloud.Dynamic 
    



      
Dynamic configuration management for Cloud modules.
This module provides functions for managing runtime configuration changes,
allowing Cloud modules to switch between different adapters and configurations
on a per-process basis, similar to Ecto's dynamic repository functionality.

  
    
  
  Usage


These functions are automatically available on Cloud modules that use
Buckets.Cloud. You typically don't need to call this module directly.

  
    
  
  Process-Local Configuration


MyApp.Cloud.put_dynamic_config([
  adapter: Buckets.Adapters.GCS,
  bucket: "tenant-bucket",
  service_account_credentials: credentials
])

  
    
  
  Scoped Configuration


MyApp.Cloud.with_config(config, fn ->
  MyApp.Cloud.insert("file.pdf")
end)

  
    
  
  Auth Server Management


For adapters that need background processes (like GCS auth servers), these
are automatically started and cached per-process as